From Derive to Maxima, the next generation

Table of contents

Introduction

The software Derive was one of the first computer algebra systems, developed by SoftWarehouse. The software was discontinued in 2007, and part of its code was used in the development of TI-Nspire CAS.

The book "Exploring Math from Algebra to Calculus with Derive, a Mathematical Assistant" by Jerry Glynn, published by MathWare in 1989, provides a very practical presentation of Derive's use in various fields of mathematics.

This document revisits its examples to show

- either the equivalent commands using Maxima
- or the procedures or programs allowing one to solve the exercises from this book.

Structure of the questions and answers from Maxima in this document:

1. first, the question or problem to be solved (next to the pen icon)
2. then, the Maxima command to be used
3. optionally, a comment appears if necessary
4. below the frame, preceded by %o, the Maxima output is displayed.

The Maxima commands should be entered exactly as indicated. However, with the recommended interface WxMaxima, most of these can be accessed through the software's menus and contextual dialog boxes.
The version used for this document is Maxima 5.46 and that of WxMaxima is 25.04. This document, as well as the PDF and *wxMaxima* versions of this file, are available on the website https://maxima-french-doc.fr.

1 Why we like Maxima as Derive's successor

1.1 Quick tour of common commands

Factorize $x^2 - 12x + 35$

```
factor(x^2-12*x+35);
```

%o1 : $(x-7)\,(x-5)$

Factorize 123454321 **into prime factors**

```
factor(123454321);
```

%o2 : $41^2\,271^2$

Solve the equation $x^2 - 23x + 132 = 0$

```
solve(x^2-23*x+132=0,x);
```

%o3 : $[x = 12, x = 11]$

Graphically represent $y = x$, $y = x^2$, $y = x^3$

```
wxplot2d([x,x^2,x^3],[x,-5,5],[y,-10,10],[color,red,blue,yellow])$
```

Comment: the plot2d command creates the plot in a new window while the wxplot2d command inserts the graphic inline in the document.

%o4 : Obtained graph



Expand $(y-13)(y+5)$

```
expand((y-13)*(y+5));
```

%o5 : $y^2 - 8\,y - 65$

Solve $x^6 - 1 = 0$

```
solve(x^6-1,x);
```

%o6 : $\left[x = \frac{\sqrt{3}\,i+1}{2}, x = \frac{\sqrt{3}\,i-1}{2}, x = -1, x = -\frac{\sqrt{3}\,i+1}{2}, x = -\frac{\sqrt{3}\,i-1}{2}, x = 1\right]$

Approximate values

```
makelist(1/n=float(1/n),n,23,25);
```

Comment: the commands float(x) or ev(x,numer) return approximate values of x.

%o7 : $[\frac{1}{23} = 0.04347826086956522, \frac{1}{24} = 0.04166666666666667, \frac{1}{25} = 0.04]$

1.2 Maxima, lists, and matrices

As with Derive, a list is defined between two brackets, and a matrix is generated as a list of lists. The automatic list generation is done with the command `makelist()`.

Create the list of the first powers of 2

```
makelist(2^n,n,1,12);
```

%o8 : $[2, 4, 8, 16, 32, 64, 128, 256, 512, 1024, 2048, 4096]$

Value of the cosine of multiples of 30 degrees

```
makelist(cos(30*n*%pi/180),n,0,6);
```

Comment: Maxima works by default in radians, so 30 degrees must be converted to radians.

%o9 : $\left[1, \frac{\sqrt{3}}{2}, \frac{1}{2}, 0, -\frac{1}{2}, -\frac{\sqrt{3}}{2}, -1\right]$

Binomial formula

```
binome1:makelist((a+b)^n=expand((a+b)^n),n,2,5);
```

%o10 : Maxima responses

$(b+a)^2 = b^2 + 2\,a\,b + a^2$
$(b+a)^3 = b^3 + 3\,a\,b^2 + 3\,a^2\,b + a^3$
$(b+a)^4 = b^4 + 4\,a\,b^3 + 6\,a^2\,b^2 + 4\,a^3\,b + a^4$
$(b+a)^5 = b^5 + 5\,a\,b^4 + 10\,a^2\,b^3 + 10\,a^3\,b^2 + 5\,a^4\,b + a^5$

Define and operate on matrices

```
matA:matrix([1,2],[4,5]);matB:matrix([-1,0],[2,3]);
matA+matB;
matA.matB;
```

Comment: use the command matrix() to define a matrix. Matrix multiplication is written with a dot.

%o11 : matA= $\begin{pmatrix}1 & 2\\ 4 & 5\end{pmatrix}$ and matB= $\begin{pmatrix}-1 & 0\\ 2 & 3\end{pmatrix}$

matA+matB= $\begin{pmatrix}0 & 2\\ 6 & 8\end{pmatrix}$ and matA.matB= $\begin{pmatrix}3 & 6\\ 6 & 15\end{pmatrix}$

1.3 Family of curves

Parametric curves

```
famille1:makelist([parametric,cos(t)^n,sin(t)^n,[t,0,2*%pi]],n,1,5)$
wxplot2d(famille1);
```

Comment: The interval for t is generated with the makelist command at the same time so that the result can be passed as a parameter to the plot2d command.

%o12 : Obtained graph



A family of lines

```
famille2:makelist(n * x, n, -4, 4, 1/4);
wxplot2d(
```

```
      famille2,
     [x,-5,5],[y,-5,5],[legend,""]);
```

%o13 : Obtained graph



1.4 Investigating Mathematics Using Maxima

1.4.1 Finding a Pattern in a Factorization

Factorize $x^n - 1$

```
facp:makelist(x^n-1=factor(x^n-1),n,2,4);
```

Comment: we are looking for a repetition in these factorizations.

%o14 : List of the first factorizations:

$x^2 - 1 = (x - 1)\ (x + 1)$
$x^3 - 1 = (x - 1)\ \left(x^2 + x + 1\right)$
$x^4 - 1 = (x - 1)\ (x + 1)\ \left(x^2 + 1\right)$
We are thus led to look more closely at $x^{2^n} - 1$:

Factorize

```
facpp:makelist(x^(2^n)-1=factor(x^(2^n)-1),n,2,4);
```

%o15 : We do find a "pattern" for these factorizations, which remains to be proven of course.

$x^4 - 1 = (x - 1)\ (x + 1)\ \left(x^2 + 1\right)$
$x^8 - 1 = (x - 1)\ (x + 1)\ \left(x^2 + 1\right)\ \left(x^4 + 1\right)$
$x^{16} - 1 = (x - 1)\ (x + 1)\ \left(x^2 + 1\right)\ \left(x^4 + 1\right)\ \left(x^8 + 1\right)$

1.4.2 Using the Definition with the Limit to Find a Derivative

Derivative of the sine function

```
delta(x,h):=(sin(x+h)-sin(x))/(x+h-x);
limit(delta(x,h),h,0);
```

%o16 : $\cos x$

1.4.3 Solving an Equation Manually

Solving manually $3x+5=21$

```
eq:3*x+5=21;
eq-5;
(eq-5)/3;
```

%o17 : $3\,x+5=21 \Longleftrightarrow 3\,x=16 \Longleftrightarrow x=\frac{16}{3}$

1.4.4 Factoring with Radicals or Complex Numbers

Maxima does not have a direct command to factor, for example, x^4+1 with radicals, unlike Derive which provides:

#1: FACTOR(x^4 + 1, Radical, x)

#2: $(x^2+\sqrt{2}\cdot x+1)\cdot(x^2-\sqrt{2}\cdot x+1)$

To obtain this result with Maxima:

```
roots: solve(x^4 + 1 = 0, x);
r1: rhs(roots[1]);
r2: rhs(roots[2]);
r3: rhs(roots[3]);
r4: rhs(roots[4]);
r1b:rectform(r1);
r2b:rectform(r2);
r3b:rectform(r3);
r4b:rectform(r4);
x^4+1=expand((x-r1b)*(x-r2b))*expand((x-r3b)*(x-r4b));
```

Comment: We solve in $\mathbb{C}$ then isolate the 4 roots which we convert to algebraic form and group two by two.

%o18 : $r1=(-1)^{\frac{1}{4}}\,i=\frac{i}{\sqrt{2}}-\frac{1}{\sqrt{2}}$ $\quad r2=-(-1)^{\frac{1}{4}}=-\frac{i}{\sqrt{2}}-\frac{1}{\sqrt{2}}$

$r3=-(-1)^{\frac{1}{4}}\,i=\frac{1}{\sqrt{2}}-\frac{i}{\sqrt{2}}$ $\quad r4=(-1)^{\frac{1}{4}}=\frac{i}{\sqrt{2}}+\frac{1}{\sqrt{2}}$

$x^4+1=\left(x^2-\sqrt{2}\,x+1\right)\,\left(x^2+\sqrt{2}\,x+1\right)$

Complex factorization with Derive:

$$\left(x+\frac{\sqrt{2}}{2}+\frac{\sqrt{2}\cdot i}{2}\right)\cdot\left(x+\frac{\sqrt{2}}{2}-\frac{\sqrt{2}\cdot i}{2}\right)\cdot\left(x-\frac{\sqrt{2}}{2}+\frac{\sqrt{2}\cdot i}{2}\right)\cdot\left(x-\frac{\sqrt{2}}{2}-\frac{\sqrt{2}\cdot i}{2}\right)$$

Maxima provides a less developed result, obtained with the command `gfactor` :

Complex factorization

```
gfactor(x^4+1);
```

%o19 : $(x^2 - i)\ (x^2 + i)$

In conclusion, Derive is always more efficient than Maxima on this question.

1.4.5 Calculating an Algebraic Form

Algebraic form of $(1+i)^{13}$

```
rectform((1+%i)^13);
```

%o20 : $-64\,i - 64$

1.4.6 Solving an Equation Graphically and Numerically

We seek to solve the equation $\cos(x) = \sin(x)$ on $[-\pi, \pi]$:

Plotting the two curves:

```
wxdraw2d(
 title="Graphic solving ",
grid=true,
xaxis=true,
yaxis=true,
 xrange=[-%pi,%pi],
 yrange=[-1.5,1.5],
    xtics=0.5,
    ytics=0.2,
color=black,
line_width=1,
key="sin(x)",
explicit(sin(x),x,-%pi,%pi),
    color=blue,
    key="cos(x)",
    explicit(cos(x),x,-%pi,%pi)
)
;
```

%o21 :



Numerical resolution

```
find_root(cos(x)=sin(x),x,-3,-2);
find_root(cos(x)=sin(x),x,0,2);
```

Comment: Maxima cannot solve this equation symbolically; it must be manually transformed into $\tan(x) = 1$.

%o22 : -2.356194490192345 and 0.7853981633974483

1.4.7 Limit Calculation

Limit of $\frac{\sin(x)}{x}$ at 0

```
limit(sin(x)/x, x, 0);
```

%o23 : 1

1.4.8 Factoring with Two Variables

Factorize $64x^6 - 125y^3$

```
factor(64*x^6-125*y^3);
```

%o24 : $-\left(5\,y-4\,x^2\right)\,\left(25\,y^2+20\,x^2\,y+16\,x^4\right)$

1.4.9 A Bit of Analytic Geometry

Define the distance between 2 points and calculate it between $(17, 11)$ and $(-5, -13)$

```
distance(A,B):=sqrt((B[1]-A[1])^2+(B[2]-A[2])^2);
distance([17,11],[-5,-13]);
```

%o25 : $\mathrm{distance}(A, B) = \sqrt{(B_2 - A_2)^2 + (B_1 - A_1)^2}$ and $\mathrm{distance}((17, 11), (-5, -13)) = 2\sqrt{265}$

Find the slope of $2x - 3y = 13$

```
expand(solve(2*x-3*y=13,y));
```

Comment: allows to find the coefficient of x, which is $\frac{2}{3}$

%o26 : $\left[y = \frac{2\,x}{3} - \frac{13}{3}\right]$

1.4.10 Representation of a Circle

With Derive, to represent the circle with equation $x^2 + y^2 = 1$, you need to draw the two semi-circles with equations $y = \pm\sqrt{1 - x^2}$. Maxima allows you to directly draw so-called implicit functions:

Plotting the circle using the implicit command from the draw package

```
wxdraw2d(
grid=true,
xaxis=true,
yaxis=true,
implicit(x^2+y^2=1,x,-2,2,y,-2,2)
)$
```



%o27 :

1.4.11 Sum Calculation

Calculate the sum of integers from 1 to 30, then from 1 to n

```
sum(k,k,1,30);
sum(k,k,1,n),simpsum;
```

%o28 : $\sum_{k=1}^{30} k = 465$ and $\sum_{k=1}^{n} k = \frac{n^2+n}{2}$

1.4.12 Area Calculation

Calculate the area between $y = x^3$ and $y = 0$ over $[0, 1]$.

```
integrate(x^3, x, 0, 1);
```

Comment: A graphical representation has been added.

%o29 : $\int_0^1 x^3 dx = \frac{1}{4}$



wxMaxima code for the graph:

```
wxdraw2d(
        grid=true,
```

```
xaxis=true,
yaxis=true,
xrange=[-1,2],
yrange=[-1.5,2],
title="Area defined by a curve",
color=black,
fill_color=red,
filled_func=x^3,
explicit(0,x,0,1),
filled_func=false,
key="y=x^3",
explicit(x^3,x,-1,2)
);
```

1.4.13 Transformation of Trigonometric Expressions

Express $\sin(6x)$ in terms of $\cos(x)$ and $\sin(x)$

```
trigexpand(sin(6*x));
trigsimp(trigexpand(sin(6*x)));
```

Comment: It is sometimes necessary to combine trigonometric simplification commands.

%o30 : $\sin(6x) = 6\cos x\sin^5 x - 20\cos^3 x\sin^3 x + 6\cos^5 x\sin x$, then with the second command: $\sin(6x) = \left(32\cos^5 x - 32\cos^3 x + 6\cos x\right)\sin x$

1.4.14 Function Composition

To obtain $f \circ g$, it is sufficient to compute $f(g(x))$ for x in the domain of definition.
To compose f n times, we can write a procedure:

For $f(x) = 1 + \frac{1}{x}$, calculate $f^{(5)}$, then its value at 1.

```
f(x):=1+1/x;
compose_n_fois(f, n) := block(
    [result: f(x)],
    for i: 1 thru n-1 do (
        result: f(result)
    ),
    define(composee(x),result),
    return(composee(x))
);
compose_n_fois(f,5);
composee(1);ev(composee(1),numer);
```

Comment: compose_n_fois takes as input the function and the number of compositions, and returns the composed function composee(x) as a result.

%o31 : $f^{(5)}(x) = \frac{1}{\frac{1}{\frac{1}{\frac{1}{\frac{1}{x}+1}+1}+1}+1} + 1$ and $f^{(5)}(1) = \frac{13}{8} \simeq 1.625$

1.4.15 And a Bit of Integration to Conclude the Chapter

Calculate $\int \ln(x)^n dx$ for n ranging from 1 to 4.

```
listeint:makelist(log(x)^i,i,1,4);
integrate(listeint,x);
```

%o32 : $\left[\log x, (\log x)^2, (\log x)^3, (\log x)^4\right]$ then

$x\ \log x - x$, $x\ \left((\log x)^2 - 2\ \log x + 2\right)$, $x\ \left((\log x)^3 - 3\ (\log x)^2 + 6\ \log x - 6\right)$ and finally $x\ \left((\log x)^4 - 4\ (\log x)^3 + 12\ (\log x)^2 - 24\ \log x + 24\right)$

2 Calculations with Maxima

Regarding numerical calculations, Maxima and Derive have the same functionalities. There is no need to detail these calculations with Maxima, as numerical operations are very straightforward with both software. The topics covered in the book "Exploring Math with Derive" include:

- How to use Derive and thus Maxima to calculate the product of two numbers mentally using decompositions and find mental calculation rules.
- Experimenting with fractions and decimals. Which fractions are decimal? Use Derive or Maxima to obtain as many decimal places as desired.

WxMaxima has dedicated commands for working with numbers, available under the Numerical menu:



Increase the default number of decimals in Maxima

```
fpprec : 100$
bfloat(%pi);
```

Comment: The fpprec command is used to increase the desired number of decimals.

%o33 : $\pi \approx 3.1415926535897932384626433832795028841971693993751b0$

3 Graphics

Derive allows plotting Cartesian or parametric curves. On this point, Maxima proves to be far more powerful than Derive. Indeed, the commands for plotting Cartesian or parametric functions are highly advanced, and it is possible to add many graphical elements (legend, title, hatching, points, vectors, etc.). The `plot2d` commands, as well as the `draw2d` commands from the draw package, are numerous and feature-rich. The examples of curve plotting presented in the book on Derive are not repeated in this chapter with Maxima, as these are basic commands for plotting curves in Maxima.

For more details, simply refer to the **Guide to Graphics with Maxima** (Maxima by Example collection), available for download on the maxima-french-doc.fr website.

As an application, the book on Derive suggests graphical research for curve intersections, root localization, etc.

4 Factorizations and Expansions

Derive and its successor Maxima have similar functionalities. The Maxima command for factorization is `factor` and for expansion is `expand`. With Derive, you must first enter the expression and then apply the factor command. With Maxima, both operations can be performed simultaneously. Here are some illustrations based on examples from the book on Derive:

Factorization examples

```
factor(7*x^4+5*x^3-3*x^2);
factor(11*x^7+33*x^3-22*x^2);
factor(28*x^4*y^2-7*x*y+21*x^3*y^2);
factor(x^2+5*x+6);
factor(x^2-7*x+5);
```

%o34 : $7\,x^4+5\,x^3-3\,x^2=x^2\,\left(7\,x^2+5\,x-3\right)$

$11\,x^7+33\,x^3-22\,x^2=11\,x^2\,\left(x^5+3\,x-2\right)$
$28\,x^4\,y^2+21\,x^3\,y^2-7\,x\,y=7\,x\,y\,\left(4\,x^3\,y+3\,x^2\,y-1\right)$
$x^2+5\,x+6=(x+2)\,(x+3)$
$x^2-7\,x+5=x^2-7\,x+5$
Maxima does not directly factor the last polynomial with roots involving radicals. Therefore, a small program extends Maxima's functionality in this regard:

Factorizations with radicals or complex numbers

```
factorbis(expression):=block(
[deg,a,roots],
    deg:hipow(expression,x),
    a:coeff(expression,x,deg),
    roots:solve(expression=0,x),
    factor_form: a*product((x - rhs(roots[i])), i, 1, length(roots)),
    return(factor_form)
);
factorbis(x^2-7*x+5);
factorbis(4*x^2-20*x+28);
```

Comment : On these examples, the command `gfactor` does not work, because by default it only factors over known fields (e.g., $\mathbb{C}$).

%o35 : $x^2-7\,x+5=\left(x+\frac{\sqrt{29}-7}{2}\right)\,\left(x-\frac{\sqrt{29}+7}{2}\right)$

$4\,x^2-20\,x+28=4\,\left(x+\frac{\sqrt{3}\,i-5}{2}\right)\,\left(x-\frac{\sqrt{3}\,i+5}{2}\right)$

An example of expansion

```
expand((x-3)^2-2);
```

%o36 : $(x-3)^2-2=x^2-6\,x+7$

Searching for a factorization rule for x^4+2^n

```
recherchefac:makelist(print("pour  n=",n,"->",factor(x^4+2^n)),
n,1,10)$;
```

Comment : This polynomial factors for x^4+2^{4n-2} (which remains to be demonstrated, of course).

%o37 :

$$\begin{bmatrix} "n=" & 1 & x^4+2 \\ "n=" & 2 & (x^2-2x+2)\ (x^2+2x+2) \\ "n=" & 3 & x^4+8 \\ "n=" & 4 & x^4+16 \\ "n=" & 5 & x^4+32 \\ "n=" & 6 & (x^2-4x+8)\ (x^2+4x+8) \\ "n=" & 7 & x^4+128 \\ "n=" & 8 & x^4+256 \\ "n=" & 9 & x^4+512 \\ "n=" & 10 & (x^2-8x+32)\ (x^2+8x+32) \end{bmatrix}$$

5 Sums

Derive and its successor Maxima easily compute sums of terms, and they do so in a very similar way:

Calculating $\sum_{n=1}^{100} n$ and $\sum_{n=1}^{k} n$

```
sum(n,n,1,100);
sum(n,n,1,k),simpsum;
```

Comment : The addition of `simsum` allows for the symbolic evaluation of the sum.

%o38 : $\sum_{n=1}^{100} n = 5050$ and $\sum_{n=1}^{k} n = \frac{k^2+k}{2}$

Additional sums: $\sum_{n=1}^{k} \frac{1}{2^n}$, $\sum_{n=1}^{100} n^7$, and $\sum_{n=1}^{k} n^7$

```
sum(1/2^n,n,1,k),simpsum;
expand(%);
sum(n^7,n,1,100);
sum(1/2^n,n,1,k),simpsum;
factor(%);
```

Comment : Derive and Maxima can correctly compute all these sums, whether they are numerical or symbolic.

%o39 : $\sum_{n=1}^{k} \frac{1}{2^n} = -2\ \left(2^{-k-1} - \frac{1}{2}\right) = 1 - \frac{1}{2^k}$

$\sum_{n=1}^{100} n^7 = 1300583304167500$ and

$\sum_{n=1}^{k} n^7 = \frac{3\,k^8+12\,k^7+14\,k^6-7\,k^4+2\,k^2}{24} = \frac{k^2\,(k+1)^2\,\left(3\,k^4+6\,k^3-k^2-4\,k+2\right)}{24}$

6 Linear Equations

6.1 Single-Variable Linear Equation

Maxima, like Derive, can solve or manipulate linear equations. In the following example, we show how to solve an equation, test a value, and manually manipulate the equation step-by-step to obtain the solution. The proposed example is $127x - 315 + 211 - 36x + 2x = 600 + 5x$.

$127x - 315 + 211 - 36x + 2x = 600 + 5x$

```
eq:127*x-315+211-36*x+2*x=600+5*x;
solve(eq,x);
subst(3,x,eq);
eq1:eq-5*x;
eq2:eq1+104;
eq2/88;
```

Comment : The `subst()` command allows testing whether 3 is a solution.

%o40 : $eq : 93\,x - 104 = 5\,x + 600$. Maxima automatically simplified the equation.

Solution found by Maxima: $[x = 8]$
Testing the value 3 in the equation: $175 = 615$. Since both sides are different, 3 is not a solution.
Subtract $5x$ from each side of the equation: $88\,x - 104 = 600$, then add 104: $88\,x = 704$
Divide by 88 to finish: $x = 8$

6.2 Linear Systems

Entering a linear system can be done via a dialog box in wxMaxima (first specify the number of equations), then enter the equations and variables. Let's solve the system (S): $\begin{cases} x + y = 27 \\ x - y = 51 \end{cases}$



Solving system (S)

```
linsolve([x+y=27, x-y=51], [x,y]);
```

Comment : User-friendliness for entering a system is improved with wxMaxima compared to Derive, where equations are entered inline.

%o41 : $[x = 39, y = -12]$

Solving a formal system $ax + by = c, dx + ey = f$

```
linsolve([a*x+b*y=c,d*x+e*y=f],[x,y]);
```

%o42 : $\left[x = -\frac{c\,e - b\,f}{b\,d - a\,e}, y = \frac{c\,d - a\,f}{b\,d - a\,e}\right]$

7 Multiple Methods for Solving Quadratic Equations

The author of the manual "Exploring Math with Derive" proposes six approaches to solve the equation $x^2 - 5x + 6 = 0$. Let's revisit them using the power of Maxima:

7.1 Graphical Method 1

From $x^2 - 5x + 6 = 0$, we deduce $x = \frac{5x-6}{x}$ by adding $5x - 6$ to the equation and then dividing by x. Substituting this value of x back into the fraction yields:

$$\frac{5x-6}{x} = \frac{5\frac{5x-6}{x} - 6}{\frac{5x-6}{x}} \Longleftrightarrow \frac{5x-6}{x} = \frac{19x-30}{5x-6}$$

We then graphically find the intersection of the two curves corresponding to the right and left sides of the equation.

Graphical solution of the quadratic equation

```
quadra1:x^2-5*x+6=0;
quadra2:quadra1+5*x-6;
quadra3:quadra2/x;
quadra4:rhs(quadra3);
define(f1(x),quadra4);
f1(quadra4);
quadra5:ratsimp(%);
define(f2(x),quadra5);
wxdraw2d(
 title="Graphical Method",
grid=true,
xaxis=true,
yaxis=true,
 xrange=[-2,5],
 yrange=[-2,10],
color=black,
line_width=1,
key="f1",
explicit(f1(x),x,-2,5),
    color=blue,
    key="f2",
    explicit(f2(x),x,-2,5)
);
```

Comment : The two functions f_1 and f_2 are defined by letting Maxima perform the calculations, then the plot of the two curves is made to read the x-coordinates of the intersection points, which gives 2 and 3 as solutions to the equation.

%o43 : quadra1: $x^2 - 5\,x + 6 = 0$, quadra2: $x^2 = 5\,x - 6$, quadra3: $x = \frac{5\,x-6}{x}$, and quadra4: $\frac{5\,x-6}{x}$

$f1(x) = \frac{5\,x-6}{x}$, and $f2(x) = \frac{19\,x-30}{5\,x-6}$



7.2 Graphical Method 2

In a much more classic way, we look for the x-intercepts of the curve defined by $y = x^2 - 5x + 6$.

Intersection with the x-axis

```
fquadra(x):=x^2-5*x+6;
wxdraw2d(
 title="Graphical Method 2",
grid=true,
xaxis=true,
yaxis=true,
 xrange=[-1,5],
 yrange=[-1,5],
color=black,
line_width=1,
key="fquadra",
explicit(fquadra(x),x,-1,5),
    color=blue,
    key="y=0",
    explicit(0,x,-1,5)
);
```

%o44 : This confirms the solutions 2 and 3.



7.3 Recurrent Sequence Approach

We use the previously obtained equality $x = 5 - \frac{6}{x}$ and define a recurrent sequence u_n by:

$$u_0 = 11 \text{ and } u_{n+1} = f(u_n) \text{ with } f(x) = 5 - \frac{6}{x}$$

The mathematical study of this sequence shows that it converges to 3. Other values for u_0 can be tested to observe the behavior of the sequence.
Note that only the value $u_0 = 2$ leads to convergence towards the other solution (value 2) of the quadratic equation.

Recurrent sequence approach

```
f(x):=5-6/x;
u(n) := block([val:11.0], /* initial value */
   for k:1 thru n do val : f(val),
   return(val)
)$
makelist(print("k=",k,"gives",u(k)),k,1,8)$
```

Comment: The initial value u_0 is defined as 11.0 to force Maxima to compute the numerical values of u_n directly.

%o45 : $u_0 = 11.0$, $u_1 = 4.454545454545455$, $u_2 = 3.653061224489796$, $u_3 = 3.357541899441341$,

$u_4 = 3.212978369384359$, $u_5 = 3.132573795960642$, $u_6 = 3.084642089601587$
This recurrent sequence can be visualized (see the code in the wxMaxima file):





7.4 Using the Maxima solve Function

The solve command

```
solve(x^2-5*x+6=0,x);
```

%o46 : $[x = 3, x = 2]$

7.5 Solving by Factorization

Solving using factorization

```
quadfac:factor(x^2-5*x+6);
facteurs : args(quadfac);
solve(facteurs[1]=0,x);
solve(facteurs[2]=0,x);
```

Comment : The `args` command converts the product of factors into a list containing each factor, which allows them to be identified with `factors[i]`.

%o47 : quadfac= $(x - 3)\ (x - 2)$

Then facteurs= $[x - 3, x - 2]$.
The respective solve commands naturally yield 3 and 2.

8 Complex Numbers

The author of "Exploring Math with Derive" introduces i and its properties. From this chapter, we only revisit the exploration of the powers of i and $(1 + i)^n$, noting that the notation for i in Maxima is %i.

Exploring the properties of i

```
makelist(%i^k,k,1,12);
makelist(expand((1+%i)^k),k,1,12);
```

%o48 : Maxima's responses for i^n: $[i, -1, -i, 1, i, -1, -i, 1, i, -1, -i, 1]$

And for $(1 + i)^n$: $[i + 1, 2\,i, 2\,i - 2, -4, -4\,i - 4, -8\,i, 8 - 8\,i, 16, 16\,i + 16, 32\,i, 32\,i - 32, -64]$
Additionally, the points with affix $(1 + i)^n$ for n ranging from 1 to 8 are plotted and connected by segments:



9 Trigonometry

Jerry Glynn uses Derive to highlight that the cosine and sine of a given angle correspond to the x-coordinate and y-coordinate of the point on the unit circle defined by that angle. This approach can be easily reproduced using Maxima's graphical functions.
Derive, and thus Maxima, can be used to obtain trigonometric identities:

Some trigonometric identities

```
cos(x+%pi);sin(%pi-x);
trigident:makelist(sin(n*x),n,1,4);
trigexpand(trigident);
```

%o49 : $\cos(x+\pi) = -\cos x$ and $\sin(\pi - x) = \sin x$

trigident: $[\sin x, \sin(2\,x), \sin(3\,x), \sin(4\,x)]$
Then the result of the transformation with trigexpand:
$[\sin x, 2\,\cos x\,\sin x, 3\,\cos^2 x\,\sin x - \sin^3 x, 4\,\cos^3 x\,\sin x - 4\,\cos x\,\sin^3 x]$

10 Representation of Trigonometric Functions

The author of *Exploring Math with Derive* uses the software to explore the representative curves of the cosine and sine functions. Thanks to the software, one can highlight the period, observe the influence of coefficients within these functions, see how the curves are modified when cosine or sine are multiplied by a given factor, and finally discover an approximation of sin(x) near 0 through a Taylor expansion. Exactly the same experiments can be carried out with Maxima. Only a few examples of these graphical explorations with Maxima are given below:

Representation of cosinus and sinus

```
wxdraw2d(
 title="Functions cos and sin",
grid=true,
xaxis=true,
yaxis=true,
 xrange=[-10,10],
 yrange=[-2,2],
color=black,
line_width=1,
```

```
key="y=cos(x)",
explicit(cos(x),x,-10,10),
    color=blue,
    key="y=sin(x)",
    explicit(sin(x),x,-10,10)
);
```

%o50 :



Horizontal shift

```
wxdraw2d(
 title="Horizontal shift of cos",
grid=true,
xaxis=true,
yaxis=true,
 xrange=[-10,10],
 yrange=[-1.5,2.5],
color=black,
line_width=1,
key="y=cos(x)",
explicit(cos(x),x,-10,10),
    color=blue,
    key="y=cos(x-1)",
    explicit(cos(x-1),x,-10,10),
 color=yellow,
    key="y=cos(x-1.5)",
    explicit(cos(x-1.5),x,-10,10)
);
```



%o51 :

Streching of cosinus

```
wxdraw2d(
 title="Streching of cosinus ",
grid=true,
xaxis=true,
yaxis=true,
 xrange=[-10,10],
 yrange=[-3,4],
color=black,
line_width=1,
key="y=cos(x)",
explicit(cos(x),x,-10,10),
    color=blue,
    key="y=1.5cos(x)",
    explicit(1.5*cos(x),x,-10,10),
 color=yellow,
    key="y=2cos(x)",
    explicit(2*cos(x),x,-10,10)
);
```



%o52 :

Approximation of sin(x) by a Taylor expansion

```
wxdraw2d(
 title="Approximation of sin by Taylor expansion",
grid=true,
xaxis=true,
yaxis=true,
 xrange=[-3,3],
 yrange=[-1.5,2],
color=black,
line_width=1,
key="y=sin(x)",
explicit(sin(x),x,-10,10),
    color=blue,
    key="y=taylor(sin(x),x,0,5)",
    explicit(taylor(sin(x),x,0,5),x,-10,10)
)
;
```

%o53 :



11 Solving Trigonometric Equations

In this chapter, Jerry Glynn focuses on the equation $\sin(x) = \frac{1}{2}$. The first part allows for an empirical study (testing values to approach a solution). The second part highlights the infinity of solutions for such a trigonometric equation. Finally, a connection is made with the formal resolution performed by the software. Let us briefly revisit these three approaches using Maxima:

Empirical approach by testing

```
makelist([0.1*i,sin(0.1*i)],i,1,6);
```

Comment : We can therefore assume, in view of the results, that there exists an x solution between 0.5 and 0.6.

%o54 : $[[0.1, 0.09983341664682816], [0.2, 0.1986693307950612], [0.3000000000000001, 0.2955202066613396]]$

$[[0.4, 0.3894183423086505], [0.5, 0.479425538604203], [0.6000000000000001, 0.5646424733950355]]$

Graphical Approach

```
wxdraw2d(
 title="Graphical approach",
grid=true,
xaxis=true,
yaxis=true,
 xrange=[-15,15],
 yrange=[-1.5,1.5],
color=black,
line_width=1,
key="y=sin(x)",
explicit(sin(x),x,-15,15),
    color=blue,
    key="y=0.5",
    explicit(0.5,x,-15,15)
);
```

Comment : This shows that the equation has an infinite number of solutions, and that thanks to the period of the sine function, it is possible to generate all the solutions once one of them is known..

%o55 :



Direct resolution with Maxima

```
solve(sin(x)=0.5,x);
```

Comment : Maxima warns that some solutions are missing.

%o56 : **(% i56)** solve(sin(x)=0.5,x);

$rat : replaced{-}0.5by{-}1/2 = -0.5solve : usingarc{-}trigfunctionstogetasolution.Somesolutionswillbelost.$

$$\left[x = \frac{\pi}{6}\right] \qquad (\%\ o56)$$

12 Some Subtleties of Maxima

12.1 Absolute Value

The absolute value function is written as `abs()` in Maxima. Maxima, just like Derive, is cautious when dealing with absolute value calculations:

Calculations of absolutes values

```
abs(5);
abs(-5);
abs(x);
assume(x>0);
abs(x);
kill(all);
assume(x<0);
abs(x);
```

Comment : The `assume()` command makes it possible to provide information about a variable. The `kill(all)` command clears the condition assigned to x; otherwise, an error occurs because x cannot be both positive and negative at the same time.

%o57 : $|5| = 5$, $|-5| = 5$ puis abs(x)= $|x|$.

Result after the command `assume(x>0);` : x Result after the command `assume(x<0);` : $-x$

12.2 Solving Inequalities

Derive can only solve a few simple inequalities. Jerry Glynn successfully tests the resolution of $2x < 12$ and $-2x < 12$, then that of $kx < 12$, which can only be solved if the sign of k is specified.
Maxima is no more efficient than Derive in this regard. By default, Maxima indicates that it cannot solve an inequality if one enters `solve(2*x<12,x)`. However, it is possible to load the package `to_poly_solve`, which allows some inequalities to be solved using the command `%solve(inequation,x)`.

Solving inequalities

```
load(to_poly_solve);
%solve(2*x<12,x);
%solve(-2*x<12,x);
```

Comment : Unlike Derive, Maxima cannot solve $kx < 12$, even if the sign of x is specified.

%o58 :

(% i7) %solve(2*x<12,x);

$$\%union\,([x<6]) \qquad (\%\ o7)$$

(% i8) %solve(-2*x<12,x);

$$\%union\,([-6<x]) \qquad (\%\ o8)$$

13 3D Graphics

Compared to Derive, Maxima's 3D graphical capabilities are much more advanced. Numerous options make it possible to control 3D scenes. Finally, the use of Gnuplot by Maxima provides significant graphical power in this context. We will give only one example for illustration purposes (the book on Derive provides others, but the process is identical).

Example of 3D Graphic

```
wxdraw3d(
  explicit(x^2-y^2, x, -3, 3, y, -3, 3),
  xlabel = "x",
  ylabel = "y",
  zlabel = "z",
  title = "Surface z = x^2 - y^2",
```

```
  xrange = [-3,3],
  yrange = [-3,3],
  zrange = [-9,9],
  color = blue
);
```

%o59 : 

14 A Bit of Calculus

14.1 The Tangent Problem

Jerry Glynn shows how to find and construct, using Derive, the tangent at the point with coordinates $(0.8; 0.6)$ to the curve defined by the equation $x^2 + y^2 = 1$. In this context, the curve is restricted to the positive half-plane. The implementation with Maxima is very straightforward thanks to the commands that make it possible to define a function and its derivative:

Curve and tangent

```
f(x):=sqrt(1-x^2);
define(derf(x),diff(f(x),x));
tangente(x0):=expand(derf(x0)*(x-x0)+f(x0));
wxdraw2d(
 title="Curve and tangent",
grid=true,
xaxis=true,
yaxis=true,
 xrange=[-2,2],
 yrange=[-1,2],
color=black,
line_width=1,
key="y=f(x)",
explicit(f(x),x,-1,1),
    color=blue,
    key="tangente en x0=0.8",
    explicit(tangente(0.8),x,-2,2)
);
```



%o60 :

The author then shows how one can approach this notion of a tangent by constructing lines passing through $(0.8; 0.6)$ and another point so as to best approximate the curve, thus highlighting the relationship between the slope of a line passing through $(0.8; 0.6)$ and the slope of the tangent.

Approach to the Tangent Using Lines

```
define_affine(x1, y1) := block(
  eq1: 0.6 = a * 0.8 + b,
  eq2: y1 = a * x1 + b,
  sol: solve([eq1, eq2], [a, b]),
  a_val: rhs(first(first(sol))),
  b_val: rhs(second(first(sol))),
  return(a_val * x + b_val)
);
droite1(x):=define_affine(0,1);
droite2(x):=define_affine(0.7,0.7);
```

Comment : Two lines are thus defined, which are added in green to the previous graph.



%o61 :

14.2 Areas

The author, Jerry Glynn, establishes a connection between an area defined by a curve, the x-axis, and the corresponding integral, inviting the reader to experiment with various examples. Maxima allows similar investigations and, in addition, provides the ability to shade an area (using the draw package).

Study of the area defined by the function sinus on $[0;\pi]$

```
wxdraw2d(
 title="Area defined by a curve",
grid=true,
xaxis=true,
yaxis=true,
 xrange=[-1,5],
 yrange=[-1.5,1.5],
color=black,
line_width=1,
key="y=sin(x)",
explicit(sin(x),x,-1,5),
    key="",
    parametric(0,t,t,-1.5,1.5),
    parametric(t,0,t,-1,5),
    fill_color=grey,
filled_func=sin(x),
explicit(0,x,0,%pi)
);
integrate(sin(x),x,0,%pi);
```

%o62 :



and we find the value of the area $\int_0^{\pi} \sin(x)\, dx = 2$
The relationship between area and integral can be easily demonstrated using the area defined by the line $y = x$ and the x-axis over the interval $[0; 1]$, which corresponds to the area of an isosceles triangle whose area is $\frac{1}{2}$. The procedure is similar to the previous example.

15 Teaching with Maxima

Jerry Glynn shows how he uses Derive for teaching mathematics. All these demonstrations can be reproduced very easily with Maxima, since the commands used are basic.